

NASA Technical Memorandum 100606

**PLOT300 - A Tektronics
PLOT10 Emulator for
HP 9000 Series 200/300
Computers**

Scott O. Kjelgaard

(NASA-TM-100606) PLOT300: A TEKTRONICS
PLOT10 EMULATOR FOR HP 9000 SERIES 200/300
COMPUTERS (NASA) 39 p CSCL 09B

N88-23447

Unclas
0142692

G3/61

May 1988



National Aeronautics and
Space Administration

Langley Research Center
Hampton, Virginia 23665-5225

PLOT300 – A Tektronics PLOT10 Emulator for HP 9000 Series 200/300 Computers

Scott O. Kjølgaard
NASA Langley Research Center
Hampton, VA 23665

ABSTRACT

A software package which emulates the Tektronics PLOT10 Graphics package on Hewlett-Packard 9000 Series 200/300 computers is described. The software is written in HP Rocky Mountain BASIC and can be run under BASIC revisions 3.0 and 4.0. Although this subroutine library emulates a subset of PLOT10, several subroutines have been added which enhance basic plot generation. Example codes using PLOT300 and descriptions of the subroutines are included in the text.

INTRODUCTION

Although the primary function of the existing NASA-Langley wind tunnel complex is the same as that of its predecessor NACA, the capabilities of the existing facilities are far superior to those of its predecessor with regard to the relative speeds obtainable, Reynolds number range, and the types and quality of data that can be acquired. The recent advances in electronics have created a wide variety of micro-processor controlled instrumentation systems and desk-top micro-computers that can interface with these systems. The capability for computer control of data acquisition systems has led to a large increase in the volume of experimental data acquired. However, this proliferation in micro-processor technology has created a wide variety of computer operating systems and languages in which the test engineer must be fluent, in order to efficiently acquire, reduce and analyze experimental data.

A typical wind-tunnel test program includes the acquisition of experimental data, real-time data reduction, to verify the quality of the acquired data, and off-line data analysis. This process can include a variety of computers, operating systems and languages. An example of this is the Basic Aerodynamics Research Tunnel at NASA-Langley Research Center. Data acquisition and real-time data display are performed by a desktop computer system using the BASIC computer language, off-line data analysis is performed on the desktop computer system, and two different mini-computers. The mini-computers both use the FORTRAN language but have entirely different operating systems. Software for displaying data on these computer systems also widely varies.

The purpose of this report is to describe the features and application of a BASIC software package which emulates the Tektronics PLOT10 graphics software package which is widely used throughout government research laboratories and industry.

EXAMPLES OF USE OF PLOT300

In the following text, examples of the use of PLOT300 are presented to show the various methods for producing a useful plot. The subroutine library PLOT300 was formulated to emulate the Tektronics PLOT10 graphics package on HP 9000 Series 200/300 computers equipped with Rocky Mountain BASIC revision 3.0 or 4.0. This software emulates a subset of the PLOT10 graphics package, but includes the subroutines required for typical engineering applications and several additional subroutines which enhance the PLOT300 graphics package. The discussion in the text assumes a fundamental knowledge of the Rocky Mountain BASIC (see references 1 thru 3) and Tektronics PLOT10 (see references 4 and 5). Descriptions of the subroutine calls and the purpose of the subroutine are presented in the next section. A listing of the PLOT300 subroutine library is presented in the appendix.

Example Code 1

This code represents the simplest method for displaying data graphically using PLOT300 and the plot produced is presented in figure 1.

```
10  OPTION BASE 1                ! Required by PLOT300
20  DIM X(100),Y(100)
30  Npoints = 11
40  FOR I = 1 TO Npoints
50    X(I) = I-1
60    Y(I) = (I-1)^2
70  NEXT I
80  CALL Initt("CRT")
90  CALL Binitt
100 CALL Npts(Npoints)
110 CALL Dsplay(X(*),Y(*))
120 END
```

Example Code 2

This sample code adds a second and third curve to the previous example plot and uses different line types and symbols to help distinguish between them. The plot produced by this code is presented in figure 2.

```
10  OPTION BASE 1                ! Required by PLOT300
20  DIM X(100), Y(100), Z(100), T(100)
30  Npoints = 11
40  FOR I = 1 TO Npoints
50    X(I) = I-1
60    Y(I) = (I-1)*7.5+10
70    Z(I) = I*10+5
80    T(I) = (I-1)*3.5+3
90  NEXT I
100 CALL Initt("CRT")
110 CALL Binitt
120 CALL Npts(Npoints)
130 CALL Syml(1)
140 CALL Line(-1)
150 CALL Dsplay(X(*),Y(*))
160 CALL Syml(2)
170 CALL Line(2)
180 CALL Npts(Npoints)
190 CALL Cplot(X(*),Z(*))
200 CALL Syml(3)
210 CALL Line(0)
220 CALL Npts(Npoints)
230 CALL Cplot(X(*),T(*))
240 END
```

Example Code 3

Example code 3 illustrates the addition of labels on the x and y axes and plot titles. The plot produced is presented in figure 3.

```
10  OPTION BASE 1                ! Required by PLOT300
20  DIM X(100), Y(100), Z(100), T(100)
```

```

30  Npoints = 11
40  FOR I = 1 TO Npoints
50    X(I) = I-1
60    Y(I) = (I-1)*7.5+10
70    Z(I) = I*10+5
80    T(I) = (I-1)*3.5+3
90  NEXT I
100 CALL Initt("CRT")
110 CALL Binitt
120 CALL Npts(Npoints)
130 CALL Syml(1)
140 CALL Line(-1)
150 CALL Hlabel("X")
160 String$ = "Y"
170 CALL Vlabel(String$)
180 CALL Title("Example code 3")
190 CALL Ptitle("Sample Plot Title")
200 CALL Dsplay(X(*),Y(*))
210 CALL Syml(2)
220 CALL Line(2)
230 CALL Npts(Npoints)
240 CALL Cplot(X(*),Z(*))
250 CALL Syml(3)
260 CALL Line(0)
270 CALL Npts(Npoints)
280 CALL Cplot(X(*),T(*))
290 END

```

Example Code 4

Changing the size of the plot and setting the minimum and maximum for the x and y axes is illustrated in example code 4. The plot produced is presented in figure 4.

```

10  OPTION BASE 1                ! Required by PLOT300
20  DIM X(100), Y(100), Z(100), T(100)
30  Npoints = 11
40  FOR I = 1 TO Npoints
50    X(I) = I-1
60    Y(I) = (I-1)*7.5+10
70    Z(I) = I*10+5
80    T(I) = (I-1)*3.5+3
90  NEXT I
100 CALL Initt("CRT")
110 CALL Binitt
120 CALL Npts(Npoints)
130 CALL Syml(1)
140 CALL Line(-1)
150 CALL Hlabel("X")
160 String$ = "Y"
170 CALL Vlabel(String$)
180 CALL Title("Example code 4")
190 CALL Ptitle("Sample Plot Title")
200 CALL Slimx(.3,.7)
210 CALL Slimy(.3,.7)

```

```

220 CALL Dlimx(-5,15)
230 CALL Dlimy(10,70,10,2)
240 CALL Dsplay(X(*),Y(*))
250 CALL Syml(2)
260 CALL Line(2)
270 CALL Npts(Npoints)
280 CALL Cplot(X(*),Z(*))
290 CALL Syml(3)
300 CALL Line(0)
310 CALL Npts(Npoints)
320 CALL Cplot(X(*),T(*))
330 END

```

Example Code 5

The addition of a legend for the lines and symbols used is presented in example code 5. The plot produced is presented in figure 5.

```

10  OPTION BASE 1                ! Required by PLOT300
20  DIM X(100), Y(100), Z(100), T(100)
30  Npoints = 11
40  FOR I = 1 TO Npoints
50    X(I) = I-1
60    Y(I) = (I-1)*7.5+10
70    Z(I) = I*10+5
80    T(I) = (I-1)*3.5+3
90  NEXT I
100 CALL Initt("CRT")
110 CALL Binitt
120 CALL Npts(Npoints)
130 CALL Syml(1)
140 CALL Line(-1)
150 CALL Hlabel("X")
160 String$ = "Y"
170 CALL Vlabel(String$)
180 CALL Title("Example code 5")
190 CALL Ptitle("Sample Plot Title")
200 CALL Slimx(.3,.7)
210 CALL Slimy(.3,.7)
220 CALL Dlimx(-5,15)
230 CALL Dlimy(10,70,10,2)
240 CALL Dsplay(X(*),Y(*))
250 CALL Syml(2)
260 CALL Line(2)
270 CALL Npts(Npoints)
280 CALL Cplot(X(*),Z(*))
290 CALL Syml(3)
300 CALL Line(0)
310 CALL Npts(Npoints)
320 CALL Cplot(X(*),T(*))
330 CALL Legend(1,1.1,.9,1,-1,1,"Line 1")
340 CALL Legend(2,1.1,.9,2,2,1,"Line 2")
350 CALL Legend(3,1.1,.9,3,0,1,"Line 3")
360 END

```

Example Code 6

Placing multiple plots on the same page is illustrated in example code 6. The plot produced is presented in figure 6.

```
10  OPTION BASE 1                ! Required by PLOT300
20  DIM X(100), Y(100), Z(100), T(100)
30  Npoints = 11
40  FOR I = 1 TO Npoints
50    X(I) = I-1
60    Y(I) = (I-1)*7.5+10
70    Z(I) = I*10+5
80    T(I) = (I-1)*3.5+3
90  NEXT I
100 CALL Initt("CRT")
110 CALL Binitt
120 CALL Npts(Npoints)
130 CALL Sybl(1)
140 CALL Line(-1)
150 CALL Place(2)
160 CALL Title("Example code 6")
170 CALL Hlabel("X")
180 CALL Vlabel("Y")
190 CALL Ptitle("Sample Plot Title")
200 CALL Dsplay(X(*),Y(*))
210 CALL Binitt
220 CALL Sybl(2)
230 CALL Line(2)
240 CALL Npts(Npoints)
250 CALL Place(6)
260 CALL Dsplay(X(*),Z(*))
270 CALL Binitt
280 CALL Sybl(3)
290 CALL Line(0)
300 CALL Npts(Npoints)
310 CALL Place(7)
320 CALL Dsplay(X(*),T(*))
330 END
```

SUMMARY OF PLOT300 SUBROUTINES

The following subroutines are available in PLOT300 and are briefly discussed in the following pages.

SUBROUTINE NAME	PAGE
AOUTST(String\$, OPTIONAL X, Y, llog, C_size)	7
BINITT	7
CPLOT(X(*), Y(*))	7
DLIMX(Xmin, Xmax, OPTIONAL X_major_tic, X_minor_tic)	8
DLIMY(Ymin, Ymax, OPTIONAL Y_major_tic, Y_minor_tic)	8
DSPLAY(X(*), Y(*))	8
HLABEL(String\$)	8
INITT(String\$)	9
LEGEND(N, X, Y, lsym, Lintyp, Ipen, String\$)	9
LINE(Lintyp)	9
NEWPAG	9
NPTS(Npoints)	10
PEN_COLOR(Pentyp)	10
PEN_DEF(Defpen)	10
PLACE(Iplace)	10
PTITLE(String\$)	10
SLIMX(X_min_gdu, X_max_gdu)	11
SLIMY(Y_min_gdu, Y_max_gdu)	11
SYMBL(lsym)	11
TITLE(String\$)	11
VLABEL(String\$)	11

DESCRIPTION OF PLOT300 SUBROUTINES

A short description of each PLOT300 subroutine follows. The PLOT10 subroutines CHECK and FINITT are not used in PLOT300.

AOUTST -

Call:

CALL AOUTST(String\$, X, Y, llog, C_size)

Parameters:

String\$ - String variable up to 15 characters lon

Optional Parameters:

X, Y - Location on plot to write String\$. Given as a fraction of the X and Y axes.

llog - Specifies relative origin of label with respect to X and Y. See LORG in reference 2.

C_size - Specifies character size to label with

Purpose:

Allows additional labels to be written on the plot. If optional parameters are not present, label is written at current pen position. If the optional parameters are present, the label is written at the X, Y supplied using LORG llog. The character size can be made larger or smaller by specifying C_size.

BINITT - See example code 1, figure 1

Call:

CALL Binitt

Parameters:

None

Purpose:

Sets constants in the PLOT300 common blocks to their default settings.

CPLOT - See example code 2, figure 2

Call:

CALL Cplot(X(*), Y(*))

Parameters:

X(*), Y(*) Arrays that contain the independent and dependent variables to be plotted

Purpose:

Allows the PLOT300 user to add additional curves to an existing plot.

DLIMX – See example code 4, figure 4

Call:

CALL Dlimx(Xmin, Xmax, OPTIONAL X_major_tic, X_minor_tic)

Parameters:

Xmin – x axis minimum

Xmax – x axis maximum

Optional Parameters:

X_major_tic – Major tic mark spacing for x axis
(major tic marks are labeled)

X_minor_tic – Minor tic mark spacing for x axis
(minor tic marks are not labeled)

Purpose:

Allows the PLOT300 user to specify the minimum and maximum values for the x axis. These values are not necessarily used as the plot maximum and minimum, but are used for target values (plotted minimum might be lower, plotted maximum might be higher). If the optional parameters are supplied, then Xmin and Xmax are used as the x axis minimum and maximum with the specified major and minor tic mark spacing.

DLIMY – See example code 4, figure 4

Call:

CALL Dlimy(Ymin, Ymax, OPTIONAL Y_major_tic, Y_minor_tic)

Parameters:

Ymin – y axis minimum

Ymax – y axis maximum

Optional Parameters:

Y_major_tic – Major tic mark spacing for y axis
(major tic marks are labeled)

Y_minor_tic – Minor tic mark spacing for y axis
(minor tic marks are not labeled)

Purpose:

Allows the PLOT300 user to specify the minimum and maximum values for the y axis. These values are not necessarily used as the plot maximum and minimum, but are used for target values (plotted minimum might be lower, plotted maximum might be higher). If the optional parameters are supplied, then Ymin and Ymax are used as the x axis minimum and maximum with the specified major and minor tic mark spacing.

DSPLAY – See example code 1, figure 1

Call:

CALL Dsplay(X(*), Y(*))

Parameters:

X(*), Y(*) – Arrays that contain the independent
and dependent variables to be plotted

Purpose:

This subroutine draws the x and y axes, frames, labels, and plots the X and Y arrays.

HLabel – See example code 3, figure 3

Call:

CALL Hlabel(String\$)

Parameters:

String\$ – String variable up to 10 characters long

Purpose:

This subroutine allows the PLOT300 user to specify the horizontal axis label written by DSPLAY

INITT - See example code 1, figure 1

Call:

CALL Init(String\$)

Parameters:

String\$ - String variable specifying plot device

Purpose:

This subroutine allows the PLOT300 user to specify the plot device and also sets default plot limits

Values for String\$ are:

"CRT" - plot device is CRT

"7475" - plot device is HPGL plotter
located at HPIB address 705

Note: If the CRT is monochromatic, the line 110 will need to be changed to PLOTTER IS CRT. If the HPGL plotter is located at another HPIB address then lines 70 and 910 will need to be modified to the correct HPIB address.

LEGEND - See example code 5, figure 5

Call:

CALL Legend(N, X, Y, Isym, Lintyp, Pen, Legen\$)

Parameters:

N - Line number of the legend

X, Y - Position of legend. Giveb in fraction of X, Y axes.

Isym - Specifies symbol number of legend

Lintyp - Specifies linetype of legend

Pen - Specifies pen number of legend

Legen\$ - string variable containing legend description

Purpose:

This subroutine allows the PLOT300 user to programatically create a plot legend.

LINE - See example code 2, figure 2

Call:

CALL Line(Lintyp)

Parameters:

Lintyp - Line type to be used for current curve

Values for Lintyp are:

-1 - no line

0 - solid line (default)

1 - dotted line

2 - dot dash

3 - dash

4 - long dash

Purpose:

This subroutine allows the PLOT300 user to specify the line type to be drawn between data points.

NEWPAG

Call:

CALL Newpag

Parameters:

None

Purpose:

Clears plot area without affecting common block constants

NPTS(Npoints) – See example code 1, figure 1

Call:

CALL Npts(Npoints)

Parameters:

Npoints – Number of data points in X and Y arrays

Purpose:

Specifies the number of data points to be plotted by the DSPLAY and CPLOT subroutines.

PEN_COLOR

Call:

CALL Pen_color(Pentyp)

Parameters:

Pentyp – Pen number used for curve

Purpose:

Allows the PLOT300 user to use different pen colors for different curves

PEN_DEF

Call:

CALL Pen_def(Defpen)

Parameters:

Defpen – Pen number used for axes and labels

Purpose:

Allows the PLOT300 user to use different pen colors for the plot axes and labels

PLACE – See example code 6, figures 6 and 7

Call:

CALL Place(Iplace)

Parameters:

Iplace – Variable specifying plot location

Values for Iplace are:

- 1 – default plotting area
- 2 – left half of plotting area (see figure 7a)
- 3 – right half of plotting area (see figure 7a)
- 4 – upper left quadrant of plotting area (see figure 7b)
- 5 – lower left quadrant of plotting area (see figure 7b)
- 6 – upper right quadrant of plotting area (see figure 7b)
- 7 – lower right quadrant of plotting area (see figure 7b)

Purpose:

Allows the PLOT300 user to place multiple plots on the same page

PTITLE – See example code 3, figure 3

Call:

CALL Ptitle(String\$)

Parameters:

String\$ – String variable up to 40 characters long

Purpose:

This subroutine allows the PLOT300 user to specify the plot title located below the horizontal axis.

SLIMX – See example code 5, figure 5

Call:

CALL SLIMX(X_min_gdu, X_max_gdu)

Parameters:

X_min_gdu – plot area x axis minimum

X_max_gdu – plot area x axis maximum

Purpose:

Allows the PLOT300 user to specify the plotting area. These values are given as fractions of the entire plotting area.

SLIMY – See example code 5, figure 5

Call:

CALL SLIMY(Y_min_gdu, Y_max_gdu)

Parameters:

Y_min_gdu – plot area y axis minimum

Y_max_gdu – plot area y axis maximum

Purpose:

Allows the PLOT300 user to specify the plotting area. These values are given as fractions of the entire plotting area.

SYMBL – See example code 2, figure 2

Call:

CALL Syml(Isym)

Parameters:

Isym – Symbol type to be used for current curve

Values for Lintyp are:

0 – no symbol (default)

1 – circle

2 – square

3 – diamond

4 – triangle

5 – upside-down triangle

6 – star

Purpose:

Allows the PLOT300 user to specify the symbol type to be drawn at each data point.

TITLE – See example code 3, figure 3

Call:

CALL Title(String\$)

Parameters:

String\$ – String variable up to 40 characters long

Purpose:

Allows the PLOT300 user to specify the plot title located above the plot.

VLABEL – See example code 3, figure 3

Call:

CALL Vlabel(String\$)

Parameters:

String\$ – String variable up to 10 characters long

Purpose:

Allows the PLOT300 user to specify the vertical axis label written by DISPLAY.

OTHER SUBROUTINES USED BY PLOT300

ASCALE - Calling subroutine - PFRAME

This subroutine sets the major and minor tick mark spacing.

MAX - Calling subroutine - DSPLAY

This subroutine find the largest value in an array.

MIN - Calling subroutine - DSPLAY

This subroutine find the smallest value in an array.

PFRAME - Calling subroutine - DSPLAY

This subroutine draws the plot axes and grid and labels.

SYMBL - Calling subroutine - DSPLAY, CPLOT

This subroutine draws the specified symbol at a specific point.

REFERENCES

1. BASIC 4.0 Programming Techniques for HP 9000 Series 200/300 Computers. Hewlett-Packard Manual No. 98613-90011, 1985.
2. BASIC 4.0 Language Reference for HP 9000 Series 200/300 Computers. Hewlett-Packard Manual No. 98613-90051, 1985.
3. BASIC 4.0 Graphics Techniques for HP 9000 Series 200/300 Computers. Hewlett-Packard Manual No. 98613-90031, 1985.
4. PLOT 10 Terminal Control System. Tektronics Part No. 070-5711-00, 1986.
5. PLOT 10 Advanced Graphing II. Tektronics Part No. 070-5713-00, 1986.

APPENDIX - PLOT300 LISTING

```

10      END                      ! Required by BASIC
20 Initt:SUB Initt(Plotter$)
30      OPTION BASE 1
40      COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
50      IF Plotter$="7475" THEN
60          Plotdev=1
70          PLOTTER IS 705,"HPGL"
80          GRAPHICS INPUT IS 705,"HPGL"
90      ELSE
100         Plotdev=0
110         PLOTTER IS 3,"INTERNAL";COLOR MAP
120         SET PEN 15 INTENSITY .3,.3,.3
130         GRAPHICS INPUT IS KBD,"ARROW KEYS"
140     END IF
150     X_gdu=100*MAX(1,RATIO)
160     Y_gdu=100*MAX(1,1/RATIO)
170     GCLEAR
180     IF SYSTEM$("SYSTEM ID")="9836C" THEN
190         Xhgdu=.9*X_gdu
200         Xlgdu=.25*X_gdu
210         Yhgdu=.80*Y_gdu
220         Ylgdu=.2*Y_gdu
230     ELSE
240         IF Plotdev=0 THEN PRINT CHR$(12)
250         Xhgdu=.85*X_gdu
260         Xlgdu=.25*X_gdu
270         Yhgdu=.85*Y_gdu
280         Ylgdu=.3*Y_gdu
290     END IF
300     VIEWPORT Xlgdu,Xhgdu,Ylgdu,Yhgdu
310 SUBEND
320 !*****
330 !*****
340 Binitt:SUB Binitt
350     OPTION BASE 1
360     COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
370     COM /Plot2/ Ndx,IncX,RincX,Ndy,Incy,Rincy
380     COM /Labels/ Hlabel$[10],Vlabel$[10],Ptitle$[40],Header$[40]
390     ! ZERO EVERYTHING
400     GRAPHICS ON
410     Hlabel$="      "
420     Vlabel$="      "
430     Ptitle$="          "
440     Header$="          "
450     Lintyp=100
460     Isym=0
470     Xlow=0
480     Xhigh=0
490     Ylow=0

```

```

500     Yhigh=0
510     Pentyp=1
520     Defpen=1
530     Ndx=0
540     Incx=0
550     Rincx=0
560     Ndy=0
570     Incy=0
580     Rincy=0
590     SUBEND
600     !*****
610     !*****
620 Pframe:SUB Pframe(Xmin,Xmax,Ymin,Ymax)
630     OPTION BASE 1
640     DEG
650     COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
660     COM /Plot2/ Ndx,Incx,Rincx,Ndy,Incy,Rincy
670     COM /Labels/ Hlabel$(10),Vlabel$(10),Ptitle$(40),Header$(40)
680     COM /Syms/ Xcir(17),Ycir(17),Xsq(5),Ysq(5),Xdi(5),Ydi(5),
Xtr(4),Ytr(4),Xutr(4),Yutr(4),Xst(6),Yst(6)
690     PEN Defpen
700     CLIP ON
710     Csize.init=MIN((Yhgdu-Ylgdu),(Xhgdu-Xlgdu))*0.025+1.2
720     IF Xlow=0 AND Xhigh=0 THEN GOTO 750
730     Xmin=Xlow
740     Xmax=Xhigh
750     IF Ndx=0 THEN
760         Xmin=Xmin+.01*(Xmax-Xmin)
770         Xmax=Xmax-.01*(Xmax-Xmin)
780         CALL Ascale(Xmin,Xmax,Xlow,Xhigh,Ndx,Incx,Rincx)
790     END IF
800     IF Ylow=0 AND Yhigh=0 THEN GOTO 830
810     Ymin=Ylow
820     Ymax=Yhigh
830     IF Ndy=0 THEN
840         Ymin=Ymin+.01*(Ymax-Ymin)
850         Ymax=Ymax-.01*(Ymax-Ymin)
860         CALL Ascale(Ymin,Ymax,Ylow,Yhigh,Ndy,Incy,Rincy)
870     END IF
880     LINE TYPE 1
890     WINDOW Xlow,Xhigh,Ylow,Yhigh
900     IF Plotdev=1 THEN
910         PRINTER IS 705
920         PRINT "VS10;"
930         FRAME
940         PRINT "VS30;"
950         PRINTER IS 1
960     ELSE
970         FRAME
980     END IF
990     AXES Incx/Rincx,Incy/Rincy,Xlow,Ylow,Rincx,Rincy,INT(.05*(Yhgdu-Ylgdu))
1000 ! LABEL AXES

```



```

1010      CSIZE Csize_init
1020      LORG 6
1030      CLIP OFF
1040      Ylow1=Ylow-(Yhigh-Ylow)*(.5+.025*(Yhgdu-Ylgdu))/(Yhgdu-Ylgdu)
1050      FOR J=1 TO Ndx+1
1060          X1=Xlow+(J-1)*IncX
1070          IF INT(X1)=0 THEN
1080              Maxdig=1
1090              GOTO 1130
1100          END IF
1110          Maxdig=INT(LGT(ABS(INT(X1))))+1
1120          IF X1<0 THEN Maxdig=Maxdig+1
1130          IF FRACT(IncX)=0 THEN
1140              Mindig=1
1150              GOTO 1180
1160          END IF
1170          Mindig=ABS(INT(LGT(ABS(FRACT(IncX)))))
1180          IF Maxdig+Mindig+1<6 THEN
1190              Form$=" "&RPT$("D",Maxdig)& "."&RPT$("D",Mindig)&""
1200          ELSE
1210              PRINT Maxdig,Mindig
1220              Form$="MD.2DE"
1230          END IF
1240          PENUP
1250          MOVE X1,Ylow1
1260          LABEL USING Form$,X1
1270      NEXT J
1280      REM LABEL Y AXES
1290      LORG 8
1300      Xlow1=Xlow-(Xhigh-Xlow)*(.01923*(Xhgdu-Xlgdu))/(Xhgdu-Xlgdu)
1310      FOR I=1 TO Ndy+1
1320          Y1=Ylow+(I-1)*IncY
1330          IF INT(Y1)=0 THEN
1340              Maxdig=1
1350              GOTO 1390
1360          END IF
1370          Maxdig=LGT(ABS(INT(Y1)))+1
1380          IF Y1<0 THEN Maxdig=Maxdig+1
1390          IF FRACT(IncY)=0 THEN
1400              Mindig=1
1410              GOTO 1440
1420          END IF
1430          Mindig=ABS(INT(LGT(ABS(FRACT(IncY)))))
1440          IF Maxdig+Mindig+1<6 THEN
1450              Form$=" "&RPT$("D",Maxdig)& "."&RPT$("D",Mindig)&""
1460              Y.digits=6
1470          ELSE
1480              Form$="MD.2DE"
1490              Y.digits=11
1500          END IF
1510          PENUP
1520          MOVE Xlow1,Y1
1530          LABEL USING Form$,Y1

```

```

1540     NEXT I
1550     ! X AXES TITLE
1560     Xmid=(Xhigh+Xlow)/2
1570     Y1=Ylow-(Yhigh-Ylow)*(1+.08*(Yhgdu-Ylgdu))/(Yhgdu-Ylgdu)
1580     LORG 6
1590     PENUP
1600     CSIZE 1.5*Csize_init
1610     MOVE Xmid,Y1
1620     CALL Aoutst(Hlabel$)
1630     REM LABEL Y AXES
1640     Ymid=(Yhigh+Ylow)/2
1650     X1=Xlow-(Xhigh-Xlow)*(1.875+.2*Y.digits*.09615*(Xhgdu-Xlgdu))/(Xhgdu-Xlgdu)
1660     LORG 8
1670     PENUP
1680     MOVE X1,Ymid
1690     CALL Aoutst(Vlabel$)
1700     ! PLOT TITLE
1710     Y1=Yhigh+(Yhigh-Ylow)*(2+(Yhgdu-Ylgdu)*.15)/(Yhgdu-Ylgdu)
1720     Xmid=Xlow+ (.5*(Xhgdu+Xlgdu)-Xlgdu)*(Xhigh-Xlow)/(Xhgdu-Xlgdu)
1730     LORG 6
1740     CSIZE 1.8*Csize_init
1750     PENUP
1760     MOVE Xmid,Y1
1770     CALL Aoutst(Header$)
1780     Y1=Ylow-(Yhigh-Ylow)*(.20*(Yhgdu-Ylgdu)+4.5)/(Yhgdu-Ylgdu)
1790     LORG 4
1800     CSIZE 1.6*Csize_init
1810     PENUP
1820     MOVE Xmid,Y1
1830     CALL Aoutst(Ptitle$)
1840     LINE TYPE INT(Lintyp/100),Lintyp MOD 100
1850     Rad=(.75+.011*(Yhgdu-Ylgdu))*(Yhigh-Ylow)/(Yhgdu-Ylgdu)
1860     Tratio=(Yhigh-Ylow)/(Xhigh-Xlow)*(Xhgdu-Xlgdu)/(Yhgdu-Ylgdu)
1870 Circle: Trad=Rad*.7071
1880     Narc=17
1890     FOR Arc=1 TO Narc
1900         Ang=(Arc-1)*360/(Narc-1)
1910         Xcir(Arc)=Trad*COS(Ang)/Tratio
1920         Ycir(Arc)=Trad*SIN(Ang)
1930     NEXT Arc
1940 Square: FOR Arc=1 TO 5
1950         Ang=45+(Arc-1)*90
1960         Xsq(Arc)=Rad*COS(Ang)/Tratio
1970         Ysq(Arc)=Rad*SIN(Ang)
1980     NEXT Arc
1990 Diamond:FOR Arc=1 TO 5
2000         Ang=(Arc-1)*90
2010         Xdi(Arc)=Rad*COS(Ang)/Tratio
2020         Ydi(Arc)=Rad*SIN(Ang)
2030     NEXT Arc
2040 Triangle:FOR Arc=1 TO 4
2050         Ang=90+(Arc-1)*120
2060         Xtr(Arc)=Rad*COS(Ang)/Tratio

```

```

2070      Ytr(Arc)=Rad*SIN(Ang)
2080      NEXT Arc
2090 Utriangle:FOR Arc=1 TO 4
2100      Ang=-90+(Arc-1)*120
2110      Xutr(Arc)=Rad*COS(Ang)/Tratio
2120      Yutr(Arc)=Rad*SIN(Ang)
2130      NEXT Arc
2140 Star:  FOR Arc=1 TO 6
2150      Ang=90+(Arc-1)*144
2160      Xst(Arc)=Rad*COS(Ang)/Tratio
2170      Yst(Arc)=Rad*SIN(Ang)
2180      NEXT Arc
2190      SUBEND
2200 !*****
2210 !*****
2220 Dsplay:SUB Dsplay(X_(*),Y_(*))
2230      OPTION BASE 1
2240      COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
2250      COM /Labels/ Hlabel$[10],Vlabel$[10],Ptitle$[40],Header$[40]
2260      Npts=Npoints
2270      CALL Max(X_(*),Xmax)
2280      CALL Min(X_(*),Xmin)
2290      CALL Max(Y_(*),Ymax)
2300      CALL Min(Y_(*),Ymin)
2310 ! PRINT "MIN,MAX",Xmin,Xmax,Ymin,Ymax
2320      CALL Pframe(Xmin,Xmax,Ymin,Ymax)
2330      PEN Pentyp
2340      CLIP ON
2350      J=0
2360      FOR I=1 TO Npts
2370          IF X_(I)<Xlow THEN GOTO 2470
2380          IF X_(I)>Xhigh THEN GOTO 2480
2390          J=J+1
2400          IF J=1 THEN
2410              PENUP
2420              MOVE X_(I),Y_(I)
2430          ELSE
2440              PLOT X_(I),Y_(I),-1
2450          END IF
2460          IF Isym>0 THEN CALL Symbol(X_(I),Y_(I),Isym)
2470      NEXT I
2480      PEN Defpen
2490      SUBEND
2500 !*****
2510 !*****
2520 Cplot:SUB Cplot(X_(*),Y_(*))
2530      OPTION BASE 1
2540      COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
2550      Npts=Npoints
2560      CLIP ON
2570      LINE TYPE INT(Lintyp/100),Lintyp MOD 100

```

```

2580     PEN Pentyp
2590     FOR I=1 TO Npts
2600         IF X_(I)<Xlow THEN GOTO 2700
2610         IF X_(I)>Xhigh THEN GOTO 2710
2620         J=J+1
2630         IF J=1 THEN
2640             PENUP
2650             MOVE X_(I),Y_(I)
2660         ELSE
2670             PLOT X_(I),Y_(I),-1
2680         END IF
2690         IF Isym>0 THEN CALL Symbol(X_(I),Y_(I),Isym)
2700     NEXT I
2710     LINE TYPE 1
2720     PEN Defpen
2730     SUBEND
2740     !*****
2750     !*****
2760 Symbol:SUB Symbol(A,B,Isymb1)
2770     OPTION BASE 1
2780     COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
2790     COM /Syms/ Xcir(17),Ycir(17),Xsq(5),Ysq(5),Xdi(5),Ydi(5),
Xtr(4),Ytr(4),Xutr(4),Yutr(4),Xst(6),Yst(6)
2800     LINE TYPE 1
2810     SELECT Isymb1
2820     CASE 1             !CIRCLE
2830         FOR Arc=1 TO 17
2840             IF Arc=1 THEN
2850                 MOVE Xcir(Arc)+A,Ycir(Arc)+B
2860             ELSE
2870                 PLOT Xcir(Arc)+A,Ycir(Arc)+B,-1
2880             END IF
2890         NEXT Arc
2900     CASE 2             !SQUARE
2910         FOR Arc=1 TO 5
2920             IF Arc=1 THEN
2930                 MOVE Xsq(Arc)+A,Ysq(Arc)+B
2940             ELSE
2950                 PLOT Xsq(Arc)+A,Ysq(Arc)+B,-1
2960             END IF
2970         NEXT Arc
2980     CASE 3             !DIAMOND
2990         FOR Arc=1 TO 5
3000             IF Arc=1 THEN
3010                 MOVE Xdi(Arc)+A,Ydi(Arc)+B
3020             ELSE
3030                 PLOT Xdi(Arc)+A,Ydi(Arc)+B,-1
3040             END IF
3050         NEXT Arc
3060     CASE 4             !TRIANGLE
3070         FOR Arc=1 TO 4
3080             IF Arc=1 THEN

```

```

3090         MOVE Xtr(Arc)+A,Ytr(Arc)+B
3100     ELSE
3110         PLOT Xtr(Arc)+A,Ytr(Arc)+B,-1
3120     END IF
3130 NEXT Arc
3140 CASE 5             ! UPSIDE DOWN TRIANGLE
3150     FOR Arc=1 TO 4
3160         IF Arc=1 THEN
3170             MOVE Xutr(Arc)+A,Yutr(Arc)+B
3180         ELSE
3190             PLOT Xutr(Arc)+A,Yutr(Arc)+B,-1
3200         END IF
3210     NEXT Arc
3220 CASE 6             ! STAR
3230     FOR Arc=1 TO 6
3240         IF Arc=1 THEN
3250             MOVE Xst(Arc)+A,Yst(Arc)+B
3260         ELSE
3270             PLOT Xst(Arc)+A,Yst(Arc)+B,-1
3280         END IF
3290     NEXT Arc
3300 END SELECT
3310 PENUP
3320 MOVE A,B
3330 LINE TYPE INT(Lintyp/100),Lintyp MOD 100
3340 SUBEND
3350 Max:SUB Max(A_(*),Amax)
3360     OPTION BASE 1
3370     COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
3380     Temp=SIZE(A_,1)
3390     REDIM A_(1:Npoints)
3400     Amax=MAX(A_(*))
3410     REDIM A_(1:Temp)
3420 SUBEND
3430 !*****
3440 !*****
3450 Min:SUB Min(A_(*),Amin)
3460     OPTION BASE 1
3470     COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
3480     Temp=SIZE(A_,1)
3490     REDIM A_(1:Npoints)
3500     Amin=MIN(A_(*))
3510     REDIM A_(1:Temp)
3520 SUBEND
3530 !*****
3540 !*****
3550 Ascale:SUB Ascale(Xmin,Xmax,Xlow,Xhigh,Nd,Inc,Rinc)
3560     OPTION BASE 1
3570     DATA 2,4,5,10
3580     ALLOCATE Cval(4)
3590     READ Cval(*)

```

```

3600      Ndiv=5
3610      Inc=(Xmax-Xmin)/Ndiv
3620      IF Inc=0 THEN
3630          Xmax=Xmax+.05
3640          Xmin=Xmin-.05
3650          Inc=(Xmax-Xmin)/Ndiv
3660      END IF
3670      Np=INT(LGT(ABS(Inc)))
3680      Inc=Inc*10^(-Np)
3690      J=1
3700      IF Inc>Cval(J) THEN GOTO Step1
3710      Inc=Cval(J)
3720      Rinc=Inc
3730      GOTO Step2
3740 Step1: J=J+1
3750      GOTO 3700
3760 Step2: IF Np<0 THEN GOTO Step3
3770      FOR K=1 TO Np
3780          Inc=Inc*10
3790      NEXT K
3800      GOTO Step4
3810 Step3: FOR K=1 TO -Np
3820          Inc=Inc/10
3830      NEXT K
3840 Step4: Xlow=INT(Xmin/Inc)*Inc
3850      Nd=INT((Xmax-Xlow)/Inc)+1
3860      Xhigh=Xlow+Nd*Inc
3870      DEALLOCATE Cval(*)
3880      SUBEND
3890      !*****
3900      !*****
3910 Symbl:SUB Symbl(Isymb1)
3920          OPTION BASE 1
3930          COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
3940          Isym=Isymb1
3950      SUBEND
3960      !*****
3970      !*****
3980 Npts:SUB Npts(Npts)
3990          OPTION BASE 1
4000          COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
4010          Npoints=Npts
4020      SUBEND
4030      !*****
4040      !*****
4050 Hlabel:SUB Hlabel(String$)
4060          OPTION BASE 1
4070          COM /Labels/ Hlabel${10},Vlabel${10},Ptitle${40},Header${40}
4080          Number=LEN(String$)
4090          Hlabel${1,Number}=String$
4100      SUBEND

```

```

4110 !*****
4120 !*****
4130 Vlabel:SUB Vlabel(String$)
4140     OPTION BASE 1
4150     COM /Labels/ Hlabel$[10],Vlabel$[10],Ptitle$[40],Header$[40]
4160     Number=LEN(String$)
4170     Vlabel$[1,Number]=String$
4180     SUBEND
4190 !*****
4200 !*****
4210 Ptitle:SUB Ptitle(String$)
4220     OPTION BASE 1
4230     COM /Labels/ Hlabel$[10],Vlabel$[10],Ptitle$[40],Header$[40]
4240     Number=LEN(String$)
4250     Ptitle$[1,Number]=String$
4260     SUBEND
4270 !*****
4280 !*****
4290 Title:SUB Title(String$)
4300     OPTION BASE 1
4310     COM /Labels/ Hlabel$[10],Vlabel$[10],Ptitle$[40],Header$[40]
4320     Number=LEN(String$)
4330     Header$[1,Number]=String$
4340     SUBEND
4350 !*****
4360 !*****
4370 Line:SUB Line(Linetype)
4380     OPTION BASE 1
4390     COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
4400     L=0
4410     IF Linetype=-1 THEN L=200
4420     IF Linetype=0 THEN L=100
4430     IF Linetype=1 THEN L=301
4440     IF Plotdev=0 THEN
4450         IF Linetype=2 THEN L=705
4460         IF Linetype=3 THEN L=400
4470         IF Linetype=4 THEN L=500
4480     ELSE
4490         IF Linetype=2 THEN L=705
4500         IF Linetype=3 THEN L=402
4510         IF Linetype=4 THEN L=508
4520     END IF
4530     Lintyp=L
4540     SUBEND
4550 !*****
4560 !*****
4570 Dlimx:SUB Dlimx(Xmin,Xmax,OPTIONAL Major_tic_x,Minor_tic_x)
4580     OPTION BASE 1
4590     COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
4600     COM /Plot2/ Ndx,IncX,RincX,Ndy,Incy,Rincy
4610     Xlow=Xmin

```

```

4620      Xhigh=Xmax
4630      IF NPAR=2 THEN SUBEXIT
4640      IF Minor.tic_x=0 THEN Minor.tic_x=Major.tic_x
4650      Incx=Major.tic_x
4660      Rincx=Major.tic_x/Minor.tic_x
4670      Ndx=INT((Xhigh-Xlow)/Major.tic_x)
4680      SUBEND
4690      !*****
4700      !*****
4710      Dlimy:SUB Dlimy(Ymin,Ymax,OPTIONAL Major.tic_y,Minor.tic_y)
4720          OPTION BASE 1
4730          COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
4740          COM /Plot2/ Ndx,Incx,Rincx,Ndy,Incy,Rincy
4750          Ylow=Ymin
4760          Yhigh=Ymax
4770          IF NPAR=2 THEN SUBEXIT
4780          IF Minor.tic_y=0 THEN Minor.tic_y=Major.tic_y
4790          Incy=Major.tic_y
4800          Rincy=Major.tic_y/Minor.tic_y
4810          Ndy=INT((Yhigh-Ylow)/Major.tic_y)
4820          SUBEND
4830      !*****
4840      !*****
4850      Slimx:SUB Slimx(X_low,X_high)
4860          OPTION BASE 1
4870          COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
4880          Xlgdu=X_low*MAX(1,RATIO)*100
4890          Xhgdu=X_high*MAX(1,RATIO)*100
4900          VIEWPORT Xlgdu,Xhgdu,Ylgdu,Yhgdu
4910          SUBEND
4920      !*****
4930      !*****
4940      Slimy:SUB Slimy(Y_low,Y_high)
4950          OPTION BASE 1
4960          COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
4970          Ylgdu=Y_low*MAX(1,1/RATIO)*100
4980          Yhgdu=Y_high*MAX(1,1/RATIO)*100
4990          VIEWPORT Xlgdu,Xhgdu,Ylgdu,Yhgdu
5000          SUBEND
5010      !*****
5020      !*****
5030      Aoutst:SUB Aoutst(String$,OPTIONAL X_per-cent,Y_per-cent,Ilogr, C.size)
5040          OPTION BASE 1
5050          COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
5060          PEN Defpen
5070          LINE TYPE 1
5080          CLIP OFF
5090          SELECT NPAR
5100          CASE 1

```



```

5110     LABEL TRIM$(String$)
5120     SUBEXIT
5130     CASE 3
5140         LORG 3
5150     CASE 4
5160         LORG lorg
5170     CASE 5
5180         LORG lorg
5190         Csize C_size
5200     END SELECT
5210     MOVE Xlow+X_per.cent*(Xhigh-Xlow),Ylow+Y_per.cent* (Yhigh-Ylow)
5220     LABEL TRIM$(String$)
5230     CLIP ON
5240     SUBEND
5250 !*****
5260 !*****
5270 Place:SUB Place(Iplace)
5280     OPTION BASE 1
5290     COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
5300     X_gdu=100*MAX(1,RATIO)
5310     Y_gdu=100*MAX(1,1/RATIO)
5320     SELECT Iplace
5330     CASE 1
5340         Xhgdu=.85*X_gdu
5350         Xlgdu=.25*X_gdu
5360         Yhgdu=.85*Y_gdu
5370         Ylgdu=.30*Y_gdu
5380     CASE 2
5390         Xhgdu=.45*X_gdu
5400         Xlgdu=.10*X_gdu
5410         Yhgdu=.75*Y_gdu
5420         Ylgdu=.30*Y_gdu
5430     CASE 3
5440         Xhgdu=.95*X_gdu
5450         Xlgdu=.60*X_gdu
5460         Yhgdu=.75*Y_gdu
5470         Ylgdu=.30*Y_gdu
5480     CASE 4
5490         Xhgdu=.5*X_gdu
5500         Xlgdu=.25*X_gdu
5510         Yhgdu=.80*Y_gdu
5520         Ylgdu=.60*Y_gdu
5530     CASE 5
5540         Xhgdu=.5*X_gdu
5550         Xlgdu=.25*X_gdu
5560         Yhgdu=.40*Y_gdu
5570         Ylgdu=.20*Y_gdu
5580     CASE 6
5590         Xhgdu=.9*X_gdu
5600         Xlgdu=.65*X_gdu
5610         Yhgdu=.80*Y_gdu
5620         Ylgdu=.60*Y_gdu

```

```

5630      CASE 7
5640          Xhgdu=.9*X_gdu
5650          Xlgdu=.65*X_gdu
5660          Yhgdu=.40*Y_gdu
5670          Ylgdu=.20*Y_gdu
5680      END SELECT
5690      VIEWPORT Xlgdu,Xhgdu,Ylgdu,Yhgdu
5700  SUBEND
5710  !*****
5720  !*****
5730 Newpag:SUB Newpag
5740      OPTION BASE 1
5750      GCLEAR
5760  SUBEND
5770  !*****
5780  !*****
5790 Pen_color:SUB Pen_color(Ipen)
5800      OPTION BASE 1
5810      COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
5820      Pentyp=Ipen
5830  SUBEND
5840  !*****
5850  !*****
5860 Pen_def:SUB Pen_def(Ipen)
5870      OPTION BASE 1
5880      COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
Isym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
5890      Defpen=Ipen
5900  SUBEND
5910  !*****
5920  !*****
5930 Set_pens:SUB Set_pens(Start_pen,OPTIONAL Re_set)
5940      OPTION BASE 1
5950      DIM Red(15),Green(15),Blue(15)
5960      SELECT NPAR
5970      CASE 1
5980          DATA 0 , 0, 0, 0, 0, 0, .67, 1, 1, 1, 1, 1, 1, 1
5990          READ Red(*)
6000          DATA 0 , .33, .67, 1, 1, 1, 1, 1, .82, .53, .29, 0, .33, .67, 1
6010          READ Green(*)
6020          DATA 1 , 1, 1, .67, .33, 0, 0, 0, 0, 0, 0, 0, .33, .67, 1
6030          READ Blue(*)
6040          SET PEN 0 COLOR 0,1,0
6050          FOR I=1 TO 15
6060              J=Start_pen+I
6070              IF J>15 THEN J=J-15
6080              SET PEN I INTENSITY Red(J),Green(J),Blue(J)
6090          NEXT I
6100      CASE 2
6110          PLOTTER IS 3,"INTERNAL";COLOR MAP
6120      END SELECT
6130  SUBEND

```

```

6140 Legend:SUB Legend(N,X1,Y1,Sym,Linetype,Pen,Legen$)
6150     OPTION BASE 1
6160     DIM String$(20)
6170     COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
lsym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
6180     CLIP OFF
6190     X1=Xlow+X1*(Xhigh-Xlow)
6200     Y1=Ylow+Y1*(Yhigh-Ylow)
6210     IF N=1 THEN MOVE X1,Y1
6220     Xscale=(Xhigh-Xlow)/(Xhgdu-Xlgdu)
6230     Yscale=(Yhigh-Ylow)/(Yhgdu-Ylgdu)
6240     X2=X1+5*Xscale
6250     Xp=X1-3*Xscale
6260     Xn=X1+3*Xscale
6270     IF Linetype=-1 THEN L=200
6280     IF Linetype=0 THEN L=100
6290     IF Linetype=1 THEN L=301
6300     IF Plotdev=0 THEN
6310         IF Linetype=2 THEN L=705
6320         IF Linetype=3 THEN L=400
6330         IF Linetype=4 THEN L=500
6340     ELSE
6350         IF Linetype=2 THEN L=708
6360         IF Linetype=3 THEN L=404
6370         IF Linetype=4 THEN L=508
6380     END IF
6390     Csize_init=(Yhgdu-Ylgdu)*.030+1.2
6400     CSIZE Csize_init
6410     Y2=Y1-(N-1)*3.5*Yscale
6420     PEN Pen
6430     CALL Symbol(X1,Y2,Sym)
6440     IF Line type=-1 THEN 6480
6450     LINE TYPE INT(L/100),L MOD 100
6460     MOVE Xp,Y2
6470     DRAW Xn,Y2
6480     LINE TYPE 1
6490     LORG 2
6500     PEN Defpen
6510     MOVE X2,Y2
6520     LABEL Legen$
6530     CLIP ON
6540     SUBEND
6550 Disvec:SUB Disvec(X(*),Y(*),Vx(*),Vy(*),Iscale)
6560     OPTION BASE 1
6570     COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
lsym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
6580     Npts=Npoints
6590     CALL Max(X(*),Xmax)
6600     CALL Min(X(*),Xmin)
6610     CALL Max(Y(*),Ymax)
6620     CALL Min(Y(*),Ymin)
6630     CALL Pframe(Xmin,Xmax,Ymin,Ymax)
6640     ! CALCULATE XSCALE AND YSCALE TO TAKE CARE OF

```

```

6650      ! VARYING ASPECT RATIOS
6660      Xscale=(Xhigh-Xlow)/(Xhgdu-Xlgdu)
6670      Yscale=(Yhigh-Ylow)/(Yhgdu-Ylgdu)
6680      PEN Pentyp
6690      FOR I=1 TO Npts
6700          PENUP
6710          MOVE X(I),Y(I)
6720          RPLOT Iscale*Xscale*Vx(I),Iscale* Yscale*Vy(I),-1
6730          ! DRAW ARROWHEAD
6740          PENUP
6750          R=SQR((Vx(I)*Xscale)^2+(Vy(I)*Yscale)^2)*Iscale
6760          Theta=ATN((Yscale*Vy(I))/(Xscale*Vx(I)))
6770          Xtip=X(I)+Vx(I)*Xscale*Iscale*.95
6780          Ytip=Y(I)+Vy(I)*Yscale*Iscale*.95
6790          Theta1=Theta+4.76364
6800          X1=COS(Theta1)*.07*R+Xtip
6810          Y1=SIN(Theta1)*.07*R+Ytip
6820          PENUP
6830          MOVE X1,Y1
6840          Theta2=Theta-4.76364
6850          X2=COS(Theta2)*.07*R+Xtip
6860          Y2=SIN(Theta2)*.07*R+Ytip
6870          PLOT X2,Y2,-1
6880          X3=COS(Theta)*.2*R+Xtip
6890          Y3=SIN(Theta)*.2*R+Ytip
6900          PLOT X3,Y3,-1
6910          PLOT X1,Y1,-1
6920      NEXT I
6930      PEN Defpen
6940      SUBEND
6950      !*****
6960      !*****
6970      Finitt:SUB Finitt
6980          OPTION BASE 1
6990          GRAPHICS OFF
7000          ALPHA ON
7010          PEN 0
7020      SUBEND
7030      !*****
7040      !*****
7050      SUB Polyfit(X(*),Y(*),D,Coef(*),Coef_of_det, Corr_coef,
St_err_est,OPTIONAL Fit_sym,Fit_lin,Fit_col)
7060          OPTION BASE 1
7070          COM /Plot1/ Xhgdu,Xlgdu,Yhgdu,Ylgdu,Xlow,Xhigh,Ylow,Yhigh,
lsym,Npoints,Lintyp,Plotdev,Pentyp,Defpen
7080          DIM A(13),R(7,8),T(8)
7090          N=Npoints
7100          A(1)=N
7110          FOR I=1 TO N
7120              FOR J=2 TO 2*D+1
7130                  A(J)=A(J)+X(I)^(J-1)
7140              NEXT J
7150          FOR K=1 TO D+1

```

```

7160      R(K,D+2)=T(K)+Y(I)*X(I)^(K-1)
7170      T(K)=T(K)+Y(I)*X(I)^(K-1)
7180      NEXT K
7190      T(D+2)=T(D+2)+Y(I)^2
7200      NEXT I
7210      FOR J=1 TO D+1
7220          FOR K=1 TO D+1
7230              R(J,K)=A(J+K-1)
7240          NEXT K
7250      NEXT J
7260      FOR J=1 TO D+1
7270          FOR K=J TO D+1
7280              IF R(K,J)<>0 THEN 7320
7290          NEXT K
7300          PRINT "NO UNIQUE SOLUTION"
7310          GOTO 7830
7320          FOR I=1 TO D+2
7330              S=R(J,I)
7340              R(J,I)=R(K,I)
7350              R(K,I)=S
7360          NEXT I
7370          Z=1/R(J,J)
7380          FOR I=1 TO D+2
7390              R(J,I)=Z*R(J,I)
7400          NEXT I
7410          FOR K=1 TO D+1
7420              IF K=J THEN 7470
7430              Z=-R(K,J)
7440              FOR I=1 TO D+2
7450                  R(K,I)=R(K,I)+Z*R(J,I)
7460              NEXT I
7470          NEXT K
7480      NEXT J
7490      FOR J=0 TO D
7500          Coef(J+1)=R(J+1,D+2)
7510      NEXT J
7520      P=0
7530      FOR J=2 TO D+1
7540          P=P+R(J,D+2)*(T(J)-A(J)*T(1)/N)
7550      NEXT J
7560      Q=T(D+2)-T(1)^2/N
7570      Z=Q-P
7580      I=N-D-1
7590      J=P/Q
7600      IF ABS(Z)<1.E-6 THEN Z=0
7610      Coef_of_det=J
7620      Corr_coef=SQR(J)
7630      St_err_est=SQR(Z/I)
7640      IF NPAR<8 THEN SUBEXIT
7650      ALLOCATE Af(N),Bf(N),F(D+1)
7660      FOR I=1 TO N
7670          Bf(I)=0
7680          Af(I)=X(I)

```

```

7690      F(1)=1
7700      FOR L=2 TO D+1
7710          F(L)=Af(L)^(L-1)
7720      NEXT L
7730      FOR L=1 TO D+1
7740          Bf(L)=Bf(L)+Coef(L)*F(L)
7750      NEXT L
7760  NEXT I
7770  CALL Syml(Fit_sym)
7780  CALL Line(Fit_lin)
7790  CALL Pen_color(Fit_col)
7800  CALL Npts(N)
7810  CALL Cplot(Af(*),Bf(*))
7820  DEALLOCATE Af(*),Bf(*),F(*)
7830  SUBEND

```




Figure 1. Plot produced by Example Code 1.





Figure 2. Plot produced by Example Code 2.


Example code 3



Sample Plot Title

Figure 3. Plot produced by Example Code 3.

Example code 4



Sample Plot Title

Figure 4. Plot produced by Example Code 4.

Example code 5




Figure 5. Plot produced by Example Code 5.







Figure 6. Plot produced by Example Code 6.




$Iplace = 2$




$Iplace = 3$


Figure 7a. Location and size of plots produced using Place(Iplace).




Iplace = 4



Iplace = 6



Iplace = 5



Iplace = 7

Figure 7b. Location and size of plots produced using Place(Iplace).



Report Documentation Page

1. Report No. NASA TM-100606		2. Government Accession No.		3. Recipient's Catalog No.	
4. Title and Subtitle PLOT300 - A Tektronics PLOT10 Emulator for HP 9000 Series 200/300 Computers				5. Report Date May 1988	
				6. Performing Organization Code	
7. Author(s) Scott O. Kjølgaard				8. Performing Organization Report No.	
				10. Work Unit No. 533-02-01-03	
9. Performing Organization Name and Address NASA Langley Research Center Hampton, VA 23665-5225				11. Contract or Grant No.	
				13. Type of Report and Period Covered Technical Memorandum	
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546-0001				14. Sponsoring Agency Code	
15. Supplementary Notes					
16. Abstract A software package which emulates the Tektronics PLOT10 Graphics package on Hewlett-Packard 9000 Series 200/300 computers is described. The Software is written in HP Rocky Mountain BASIC and can be run under BASIC revisions 3.0 and 4.0. Although this subroutine library emulates a subset of PLOT10, several subroutines have been added which enhance basic plot generation. Example codes using PLOT300 and descriptions of the subroutines are included in the text.					
17. Key Words (Suggested by Author(s)) BASIC, Graphics Package Line plots, PLOT10, Computer Graphics				18. Distribution Statement Unclassified - Unlimited Subject Category 61	
19. Security Classif. (of this report) Unclassified		20. Security Classif. (of this page) Unclassified		21. No. of pages 38	
				22. Price A03	